Algoritmi astronomici utilizzabili per il calcolo dell'auto-induttanza di una spira ellittica piatta di larghezza w

Giuseppe Matarazzo

Luglio 2006

Sommario

Succede spesso, nelle scienze applicate, che alcuni procedimenti di calcolo possano essere *interscambiabili*. In questo caso la Meccanica Celeste "presta" all'Elettromagnetismo le formule necessarie alla determinazione di un importante parametro, l'*auto-induttanza* L, che nel caso di una spira a corona circolare non presenta quelle difficoltà di calcolo che invece si manifestano quando la forma è ellittica.

Introduzione

Il procedimento scelto è quello derivato dalla bobina a gabbia d'uccello (*birdcage coil*) suddivisa in N tratti, generalmente in numero di 8, ad altezza H nulla. E infatti quest'ultimo parametro non comparirà negli algoritmi di calcolo.
La formula dell'*auto-induttanza* L è stata ricavata da un documento privato [1] e riguarda il calcolo di un *integrale quadruplo*

Induttanza in una spira circolare

In una spira a corona circolare l'induttanza vale:

$$L = \frac{\mu_0}{4\pi w^2} \int_0^w \int_0^{\frac{2\pi}{N}} \int_0^w \int_0^{\frac{2\pi}{N}} \frac{\cos(\theta' - \theta)(a+r)(a+r')}{R} dr dr' d\theta d\theta' \tag{1}$$

essendo R la distanza tra i due elementi *infinitesimi* di apertura $d\theta$ e $d\theta'$, che equivale a:

$$R = \sqrt{[(a+r)\cos\theta - (a+r')\cos\theta']^2 + [(a+r)\sin\theta - (a+r')\sin\theta']^2}$$

e inoltre w la larghezza della piastra ed a il raggio interno della spira circolare, mentre μ_0 è il valore della *permeabilità magnetica* nel vuoto.
La corrente elettrica I che percorre il conduttore è costante e non figura nel calcolo dell'induttanza.

[1] **"Calcolo dell'induttanza di una spira circolare"** di Giulio Giovannetti, S.Cataldo (Pisa)



Figura 1: Coordinate cartesiane dell'ellisse

Induttanza in una spira ellittica

Prendiamo in considerazione la figura 1, che è tratta dal mio lavoro [2].
Si nota, rispetto al caso della spira circolare, come i *raggi osculatori* N e N' siano variabili in funzione delle *sotto-normali* ϕ e ϕ', restando invece costante l'angolo $(\phi' - \phi)$ tra le normali ai suddetti raggi, il cui coseno interviene nella funzione integranda. Come mostrato sotto.

$$L = \frac{\mu_0}{4\pi w^2} \int_0^w \int_0^{\frac{2\pi}{N}} \int_0^w \int_0^{\frac{2\pi}{N}} \frac{N \cdot N' \cdot \cos(\phi' - \phi)}{R} dr dr' d\phi d\phi' \tag{2}$$

$$\text{con} \quad N = \frac{a}{\sqrt{1 - e^2 \cdot \sin^2\phi}} \qquad N' = \frac{a}{\sqrt{1 - e^2 \cdot \sin^2\phi'}} \qquad R = \sqrt{(x-x')^2 + (y-y')^2}$$

[2] **"Coordinate Dirette e Inverse (metodo di Borkowski) di un punto P in un sistema di riferimento Geocentrico"** - Dic.2005



Figura 2: Elementi differenziali della spira ellittica

Procedimento di calcolo

Il metodo risolutivo dell'integrale quadruplo viene effettuato suddividendo le N "campate" in $M_3 = 30$ intervalli finiti lungo la normale ai raggi vettori e in $M_4 = 6$ intervalli nella direzione di tali raggi.

La corona di spira ellittica della figura 2 è suddivisa in $N = 8$ settori, individuati dai numeri romani. L'induttanza del settore (II) si calcola sommando, con il cosiddetto metodo dei *pluri-rettangoli*, i contributi dei settori (II-III), (II-IV), ... (II-I). Successivamente si passa all'induttanza del settore (III) con i contributi di (III-IV), (III-V), ... (III-II), e, procedendo in senso antiorario si prosegue, fino a completare gli 8 settori.

L'*induttanza totale* si ottiene come *sommatoria* delle induttanze degli 8 settori. Non è detto, comunque, che i settori siano necessariamente quelli indicati in figura; infatti nel codice del programma computerizzato mostrato sotto si può scegliere qualsiasi valore di N. Naturalmente, per valori più alti di N, si possono diminuire gli intervalli finiti M_3 dei settori di corona ellittica.

Listato Basic

```
' -------- Calcolo auto-induttanza di una spira ellittica  -------------------
'                        (Integrale Quadruplo)
'
  COLOR 14, 1: CLS
  DEFDBL A-Z
  pi = 4 * ATN(1#)
'-------------------------
' Costanti del problema
 Mo = .00000126#
 cost = Mo / (4 * pi)
 M3 = 30: 'intervalli
 M4 = 6: 'intervalli
'-------------------------
'Dati in centimetri
 a=10.0
 b=8.9
 w=1.0
 Nlegs=8: 'N.ro elementi finiti
'-------------------------
'Dati in metri
 a = a / 100
 b = b / 100
 w = w / 100
'----------------
' Calcolo Eccentricita'
 e = SQR(1 - (b / a) ^ 2)
'
 LOCATE 12, 25: PRINT "Eccentricita'  e= "; e
PRINT
'
'----------------------------
'  Calcolo Integrale Quadruplo
'----------------------------
Sommaintegra1 = 0
FOR Kount = 1 TO Nlegs
 Integra1 = 0
 FOR k = 1 TO Nlegs

      alfa1 = (Kount - 1) * 2 * pi / Nlegs
      alfa2 = (Kount - 1) * 2 * pi / Nlegs + 2 * pi / Nlegs

      beta1 = k * ((2 * pi) / Nlegs)
      beta2 = (k + 1) * ((2 * pi) / Nlegs)
      larg1 = 0
      lorg1 = 0
      incalfa = (alfa2 - alfa1) / M3
      incbeta = (beta2 - beta1) / M3
'-----------------------------------------
```

```
   FOR la = 0 TO M3 - 1
      alfa = alfa1 + la * incalfa
      N1 = a / SQR(1 - e ^ 2 * SIN(alfa) ^ 2): ' raggio osculatore N1

    FOR lo = 0 TO M4 - 1
        inclarg = (w - larg1) / M4
        larg = larg1 + lo * inclarg

     FOR al = 0 TO M3 - 1
       beta = beta1 + al * incbeta
       N2 = a / SQR(1 - e ^ 2 * SIN(beta) ^ 2): ' raggio osculatore N2

      FOR be = 0 TO M4 - 1
        inclorg = (w - lorg1) / M4
        lorg = lorg1 + be * inclorg
'------------------------------------------------------
         den1 = SQR((N1 * COS(alfa) - N2 * COS(beta)) ^ 2 + (N1 * (1 - e ^ 2) *
                SIN(alfa) - N2 * (1 - e ^ 2) * SIN(beta)) ^ 2 + (larg - lorg) ^ 2)
           IF den1 > .00001 THEN
           Integra1 = Integra1 + (COS(beta - alfa) * N1 * N2 * incalfa * incbeta *
                      inclarg * inclorg) / den1
           END IF
      NEXT be
     NEXT al
    NEXT lo
   NEXT la
 NEXT k
  LOCATE CSRLIN, 25: PRINT USING "Kount=##   I-parz= #.######^^^^"; Kount; Integra1
  Sommaintegra1 = Sommaintegra1 + Integra1
 NEXT Kount

      Lsett = (cost * Sommaintegra1) / (w ^ 2)
'
PRINT
LOCATE CSRLIN, 22: COLOR 14, 3: PRINT "Auto-induttanza Spira Ellittica": COLOR 14, 1
LOCATE CSRLIN, 26: PRINT "L= "; Lsett
LOCATE CSRLIN, 26: PRINT "I= "; Sommaintegra1; : PRINT #2, "(Integrale)"
WHILE INKEY$ = "": WEND
 END

'................ RISULTATI .....................
'
 Eccentricita'  e=  .45596059906708

  Kount= 1   I-parz= 0.664696D-04
  Kount= 2   I-parz= 0.630166D-04
  Kount= 3   I-parz= 0.629197D-04
  Kount= 4   I-parz= 0.663851D-04
  Kount= 5   I-parz= 0.664696D-04
```

```
  Kount= 6   I-parz= 0.630166D-04
  Kount= 7   I-parz= 0.629197D-04
  Kount= 8   I-parz= 0.663851D-04
'
  Auto-induttanza Spira Ellittica
  L=  5.1896704844683D-07
  I=  5.17581925985949D-04 (Integrale)
```

Semplificazione del metodo

E' intuitivo che il procedimento, appunto perchè si avvale, nel calcolo della funzione integranda, della sommatoria di tantissimi rettangolini, possa essere esteso ad estremi di integrazione di $d\phi$ e $d\phi'$ variabili in modo continuo da 0 a 2π. In tal caso il numero degli intervalli M_3 va opportunamente modificato, passando dal valore $M_3 = 30$ per l'arco complessivo di $\pi/4$ a $M_3 = 240$ per tutto l'angolo giro.
Il numero totale dei *cicli* di calcolo resta comunque *invariato* e, per gli intervalli predefiniti, ammonta a 2.072.160. Il listato *semplificato*, nel senso che non occorre definire a-priori il numero N di settori di corona ellittica (come succede per la "gabbia d'uccello"), si presenta così:

```
' -------- Calcolo auto-induttanza di una spira ellittica  -------------------
'                        (Integrale Quadruplo)
' ---------------------------------------------------------------------------
'
  COLOR 14, 1: CLS
  DEFDBL A-Z
  pi = 4 * ATN(1#): rad = pi / 180
'-------------------------
'Dati in centimetri
 a=10.0
 b=8.9
 w=1.0
'-------------------------
' Costanti del problema
 Mo = .00000126#
 cost = Mo / (4 * pi)
 M3 = 240: 'intervalli   :' ex M3=30
 M4 = 6: 'intervalli
'-------------------------
'Dati in metri
 a = a / 100
 b = b / 100
 w = w / 100
'-----------------
' Calcolo Eccentricita', 2 costanti correlate e parametro (p)
 e = SQR(1 - (b / a) ^ 2)
'
```

```
 LOCATE 12, 25: PRINT "Eccentricita'  e= "; e
PRINT
'
'-------------------------------
'  Calcolo Integrale Quadruplo
'-------------------------------
Sommaintegra1 = 0
 Integra1 = 0
      alfa1 = 0
      alfa2 = 2 * pi

      beta1 = 0
      beta2 = 2 * pi
      larg1 = 0
      lorg1 = 0
      incalfa = (alfa2 - alfa1) / M3
      incbeta = (beta2 - beta1) / M3
'------------------------------------------

   FOR la = 1 TO M3
      alfa = alfa1 + la * incalfa
      N1 = a / SQR(1 - e ^ 2 * SIN(alfa) ^ 2): ' raggio osculatore N

    FOR lo = 1 TO M4
        inclarg = (w - larg1) / M4
        larg = larg1 + lo * inclarg

     FOR al = 1 TO M3
       beta = beta1 + al * incbeta
       N2 = a / SQR(1 - e ^ 2 * SIN(beta) ^ 2): ' raggio osculatore N'

      FOR be = 1 TO M4
        inclorg = (w - lorg1) / M4
        lorg = lorg1 + be * inclorg
'------------------------------------------------------
         den1 = SQR((N1 * COS(alfa) - N2 * COS(beta)) ^ 2 + (N1 * (1 - e ^ 2) *
                SIN(alfa) - N2 * (1 - e ^ 2) * SIN(beta)) ^ 2 + (larg - lorg) ^ 2)
          IF den1 > .00001 THEN
          Integra1 = Integra1 + (COS(beta - alfa) * N1 * N2 * incalfa * incbeta *
                     inclarg * inclorg) / den1
  LOCATE 13, 25: PRINT " Fi-gradi [0-360]= ";: COLOR 14, 4: PRINT USING "###.#";
                 alfa/rad: COLOR 14, 1
  LOCATE 14, 25: PRINT "Fi'-gradi [0-360]= ";: COLOR 14, 4: PRINT USING "###.#";
                 beta/rad: COLOR 14, 1
          END IF
      NEXT be
     NEXT al
    NEXT lo
   NEXT la
  Sommaintegra1 = Sommaintegra1 + Integra1
```

```
      Lsett = (cost * Sommaintegra1) / (w ^ 2)
'
PRINT
LOCATE CSRLIN, 22: COLOR 14, 3: PRINT "Auto-induttanza Spira Ellittica":COLOR 14,1
LOCATE CSRLIN, 26: PRINT "L= "; Lsett
LOCATE CSRLIN, 26: PRINT "I= "; Sommaintegra1; : PRINT "(Integrale)"

WHILE INKEY$ = "": WEND
 END

'................. RISULTATI ......................
'
 Eccentricita'  e=  .45596059906708
  Auto-induttanza Spira Ellittica
  L=  5.18967048446844D-07
  I=  5.17581925985963D-04 (Integrale)
```

Conclusione

Una *verifica sperimentale* è quindi auspicabile e si spera che questa relazione, seppur sintetica, possa suscitare l'interesse di qualche ingegnere del ramo.
In un confronto teorico del metodo proposto, possiamo constatare la perfetta rispondenza tra l'integrale dell'induttanza $I = 4.406253 \cdot 10^{-4}$ ottenuto con l'algoritmo della spira circolare (a=10 cm, w=1 cm) e l'identico $I = 4.406253 \cdot 10^{-4}$ ricavato con i suesposti programmi della spira ellittica, resa circolare ponendo a=b=10 e larghezza w=1.
(31-07-2006)